L2 Calcul formel Tp 2 : Systèmes d'équations linéaires entières et à paramètres polynomiaux

```
> restart;
```

N'oubliez pas de charger la bibliothèque 'linalg'!

```
> with(linalg):
Warning, new definition for norm
Warning, new definition for trace
```

Algorithme de réduction des matrices entières et polynomiales

Matrices entières

Tester l'algorithme présenté en cours sur les matrices entières M suivantes. Vous devez donc trouver une suite d'opérations élémentaires sur les lignes et les colonnes pour "diagonaliser" la matrice M.

Les fonctions arithmétiques `irem` et `iquo` pourront vous être utiles.
Dans Maple, les opérations élémentaires sur les lignes et les colonnes sont implémentées dans les fonctions `addrow`/`addcol` et `swaprow`/`swapcol` (en anglais, to swap=échanger).

Pour la suite, il est utile de se souvenir des opérations élémentaires effectuées, ou du moins de trouver des matrices entières inversibles g (comme gauche) et d (comme droite) telles que gMd soit diagonale.
Pour vous aider, je fournis des variantes `opC`/`opL` et `ech` qui construisent les matrices g et d à mesure que vous faîtes vos opérations élémentaires. Leur syntaxe est la suivante:
-`opC(i,r,j)` change Ci en Ci+r*Cj
-`opL(i,r,j)` change Li en Li+r*Lj.
-`ech(i0,j0,i1,j1)` effectue un échange de lignes et un échange de colonnes pour que les coefficients d'indices (i0,j0) et (i1,j1) soient échangés.
En prime, une fonction `annule()` permet de revenir en arrière si l'on s'est trompé.
Lorsque vous avez fini, la fonction `affiche(M)`, vous permet de récupérer les valeurs des matrices g et d construites et de vérifier votre résultat.
Si vous avez besoin de récupérer des coefficients dans la
Si vous le souhaitez, vous pouvez regarder rapidement comment fonctionnent

les fonctions suivantes, mais ce n'est pas indispensable.

```
> init:=proc(M)
  local n,p;
  global niveau,sauveG, sauveD, sauveM;
  n:=rowdim(M);
  p:=coldim(M);
  niveau:=1:
  sauveG[0]:=matrix(n,n, (i,j)-> if i=j then 1 else 0 fi):
  sauveD[0]:=matrix(p,p, (i,j)-> if i=j then 1 else 0 fi):
  sauveM[0]:=copy(M):
  end:

> transvL:=proc(i,j,r)
  local T;
  T:=copy(sauveG[0]);
  T[i,j]:=r;
  evalm(T);
  end:

  transvC:=proc(i,j,r)
  local T;
  T:=copy(sauveD[0]);
  T[i,j]:=r;
  evalm(T);
  end:

  permL:=proc(i,j)
  local E;
  E:=copy(sauveG[0]);
  E[i,i]:=0; E[j,j]:=0; E[i,j]:=1; E[j,i]:=1;
  E;
  end:

  permC:=proc(i,j)
  local E;
  E:=copy(sauveD[0]);
  E[i,i]:=0; E[j,j]:=0; E[i,j]:=1; E[j,i]:=1;
  E;
  end:


  opL:=proc(i,r,j)
  local R;
  global niveau,sauveM,sauveG,sauveD,N;
  R:=map(expand,addrow(sauveM[niveau-1],j,i,r));
  sauveM[niveau]:=R;
```

```
sauveD[niveau]:=sauveD[niveau-1];
sauveG[niveau]:=multiply(transvL(i,j,r),sauveG[niveau-1]);

niveau:=niveau+1;
N:=evalm(R);
end:

opC:=proc(j,r,i)
local R;
global niveau,sauveM,sauveG,sauveD,N;
R:=map(expand,addcol(sauveM[niveau-1],i,j,r));
sauveM[niveau]:=R;
sauveD[niveau]:=multiply(sauveD[niveau-1],transvC(i,j,r));

sauveG[niveau]:=sauveG[niveau-1];
niveau:=niveau+1;
N:=evalm(R);
end:

ech:=proc(i0,j0,i1,j1)
local R;
global niveau,sauveM,sauveG,sauveD,N;
R:=swaprow(swapcol(sauveM[niveau-1],j0,j1),i0,i1);
sauveM[niveau]:=R;
sauveD[niveau]:=multiply(sauveD[niveau-1],permC(j0,j1)) ;
sauveG[niveau]:=multiply(permL(i0,i1),sauveG[niveau-1]) ;
niveau:=niveau+1;
N:=evalm(R);
end:

annule:=proc()
global niveau,N;
niveau:=niveau-1:
N:=evalm(sauveM[niveau-1]);
end:

affiche:=proc(M)
global niveau,g,d;
g:=sauveG[niveau-1];
d:=sauveD[niveau-1];
print('g'=map(sort,map(expand,evalm(g))));
print('d'=map(sort,map(expand,evalm(d))));
print('gMd'=map(sort,map(expand,multiply(g,M,d))));
end:
> M := matrix([[9, -36, 30], [-36, 192, -180], [30, -180,
180]]); init(M):
```

$$M := \begin{bmatrix} 9 & -36 & 30 \\ -36 & 192 & -180 \\ 30 & -180 & 180 \end{bmatrix}$$

```
>
>
>
>
>
```

```
> M1:=matrix([[41, -58, -90, 53, -1], [94, 83, -86, 23,
-84], [19, -50, 88, -53, 85]]);
init(M1):
```

$$M1 := \begin{bmatrix} 41 & -58 & -90 & 53 & -1 \\ 94 & 83 & -86 & 23 & -84 \\ 19 & -50 & 88 & -53 & 85 \end{bmatrix}$$

```
>
```

Comparez vos résultats avec ceux de la fonction `ismith` de Maple.

Matrices polynomiales

Même question, avec des matrices à coefficients polynomiaux.
Attention: la division euclidienne dans les polynômes se fait avec les fonctions `rem` et `quo`.

```
> M3:=matrix([[6*x^8+30*x^7+33*x^6+72*x^5+28*x^4-15*x^3-48*x
^2-81*x-23,
12*x^8+40*x^7+40*x^6+56*x^5+15*x^4-38*x^3-30*x^2-63*x-32,
2*x^5+6*x^4+3*x^3+2*x^2-5*x-8],
[21*x^8+31*x^7+70*x^6+75*x^5+14*x^4-46*x^3-91*x^2-61*x-13,
42*x^8+62*x^7+63*x^6+48*x^5-20*x^4-54*x^3-69*x^2-55*x-17,
7*x^5+8*x^4+2*x^3-4*x^2-9*x-4],
[3*x^5+x^4+5*x^3+2*x^2-8*x-3, 6*x^5+2*x^4-x^3+2*x^2-5*x-4,
x^2-1]]);
init(M3):
```

```
>
```

```
>
```

Comparez vos résultats avec ceux de la fonction `smith` de Maple.

Dans le temps qu'il vous reste, essayez d'automatiser vos fonctions. L'idéal serait d'arriver à reprogrammer la fonction smith.

Solutions entières de systèmes d'équations entières

Exercice 1: un système à résoudre

Utilisez la méthode présentée en cours pour décrire l'ensemble des solutions entières du système suivant (vous avez le droit d'utiliser la fonction ismith).

```
> syst:={-50*x-76*y-79*z+23*t+40*u=0,-92*x-16*y+32*z+34*t-34*
  u=0,4*x-72*y+12*z+92*t-74*u=0};
```

$$syst := \{-50\,x - 76\,y - 79\,z + 23\,t + 40\,u = 0, -92\,x - 16\,y + 32\,z + 34\,t - 34\,u = 0, 4\,x - 72\,y + 12\,z + 92\,t - 74\,u = 0\}$$

La commande Maple permettant de résoudre des équations diophantiennes (non nécessairement linéaires) s'appelle isolve. Vérifiez que votre résultat coïncide avec celui de Maple.

Une solution

On applique la méthode présentée en cours. On écrit notre système sous forme matricielle

MX=0, puis on applique des opérations élémentaires pour écrire PMQ=D avec D une matrice "diagonale".

```
> M:=genmatrix(syst,[x,y,z,t,u]);
```

$$M := \begin{bmatrix} -50 & -76 & -79 & 23 & 40 \\ -92 & -16 & 32 & 34 & -34 \\ 4 & -72 & 12 & 92 & -74 \end{bmatrix}$$

```
> d:=ismith(M,P,Q);
```

$$d := \begin{bmatrix} 1 & 0 & 0 & 0 & 0 \\ 0 & 2 & 0 & 0 & 0 \\ 0 & 0 & 2 & 0 & 0 \end{bmatrix}$$

L'ensemble des solutions du système dY=0 a pour base :

```
> e4:=vector([0,0,0,1,0]);
  e5:=vector([0,0,0,0,1]);
```

$$e4 := [0, 0, 0, 1, 0]$$

$$e5 := [0, 0, 0, 0, 1]$$

L'ensemble des solutions de MX=0 admet alors pour base:

```
> v1:=evalm(Q &* e4);
  v2:=evalm(Q &* e5);
```

$$v1 := [7131, -31882, 45826, 4, 38842]$$

$$v2 := [8349, -37331, 53654, 0, 45474]$$

On compare avec les solutions de la commande Maple isolve.

```
> S:=isolve(syst,{a,b});
```

$$S := \{z = 2\,a, u = 47918\,a + 68572\,b, x = 1887\,a + 2700\,b, t = 74984\,a + 107308\,b, y = 46669\,a + 66789\,b\}$$

Maple indique la base suivante:

```
> assign(S);
```

```
> w1:=subs({a=1,b=0},[x,y,z,t,u]);
  w2:=subs({a=0,b=1},[x,y,z,t,u]);
```

$$w1 := [1887, 46669, 2, 74984, 47918]$$

$$w2 := [2700, 66789, 0, 107308, 68572]$$

Comme cette réponse a l'air assez différente de la nôtre, mieux vaut vérifier que les deux

réponses coïncident, c'est-à-dire que ce sont deux bases entières du même ensemble.
Cela revient à vérifier que v1 et v2 s'expriment comme combinaison linéaire **entière** de w1 et w2 et qu'inversement w1 et w2 s'expriment comme combinaison linéaire **entière** de v1 et v2.

```
> linsolve(augment(v1,v2),w1);
```

$$[18746, -16011]$$

```
> linsolve(augment(v1,v2),w2);
```

$$[26827, -22913]$$

```
> linsolve(augment(w1,w2),v1);
```

$$[22913, -16011]$$

```
> linsolve(augment(w1,w2),v2);
```

$$[26827, -18746]$$

Ouf, la solution de Maple coïncide bien avec la nôtre!

Exercice 2: base d'un réseau

Paramétrer l'ensemble des combinaisons linéaires **rationnelles** des vecteurs v1, v2, v3 qui sont entières.

```
> v1:= vector([2,2/3,1,1/2,1]);
  v2:= vector([3,1/2,0,1/4,1]);
  v3:= vector([2,1,1,4/3,1/3]);
```

$$v1 := \left[2, \frac{2}{3}, 1, \frac{1}{2}, 1\right]$$

$$v2 := \left[3, \frac{1}{2}, 0, \frac{1}{4}, 1\right]$$

$$v3 := \left[2, 1, 1, \frac{4}{3}, \frac{1}{3}\right]$$

Par exemple, le vecteur w=1/2*v1+1/3*v2+1/2*v3 est une combinaison linéaire rationnelle qui est entière.

```
> w:=evalm(1/2*v1+1/3*v2+1/2*v3);
```

$$w := [3, 1, 1, 1, 1]$$

Une solution

Il faut se ramener au thème du moment, à savoir la résolution d'un système d'équations entières!
Oui, on est juste en train de chercher les points entiers dans le Q-espace vectoriel E de base <v1,v2,v3>.
Il nous suffit donc de trouver un système d'équations pour E et de trouver les solutions entières de ce système.

```
> M:=transpose(augment(v1,v2,v3));
```

$$M := \begin{bmatrix} 2 & \frac{2}{3} & 1 & \frac{1}{2} & 1 \\ 3 & \frac{1}{2} & 0 & \frac{1}{4} & 1 \\ 2 & 1 & 1 & \frac{4}{3} & \frac{1}{3} \end{bmatrix}$$

```
> K:=op(kernel(M));
```

$$K := \left[1, \frac{-15}{2}, \frac{3}{2}, 3, 0 \right], [0, -3, 0, 2, 1]$$

Un système d'équations pour E est donc

```
> x:=vector(5);
  systE:=geneqns(matrix(1,5,K[1]),x) union
  geneqns(matrix(1,5,K[2]),x);
```

$$x := \operatorname{array}(1 .. 5, [\ ])$$

$$systE := \{ x_1 - \frac{15}{2} x_2 + \frac{3}{2} x_3 + 3\, x_4 = 0, -3\, x_2 + 2\, x_4 + x_5 = 0 \}$$

Si l'on est gêné par les dénominateurs, on peut bien multiplier la première équation par 2, cela reste un système d'équations définissant E. (Mutiplier une équation par un scalaire ne change pas l'ensemble de ses solutions, même l'ensemble de ses solutions entières!).

```
> M:=transpose(augment((2*K[1]),K[2]));
```

$$M := \begin{bmatrix} 2 & -15 & 3 & 6 & 0 \\ 0 & -3 & 0 & 2 & 1 \end{bmatrix}$$

```
> ismith(M,P,Q);
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 \end{bmatrix}$$

Comme dans l'exercice précédent, une base du noyau entier de M est donnée par les 3 dernières colonnes de Q:

```
> w1:=col(Q,3);
  w2:=col(Q,4);
  w3:=col(Q,5);
```

$$w1 := [6, 1, 1, 0, 3]$$

$$w2 := [12, 2, 0, 1, 4]$$

$$w3 := [15, 2, 0, 0, 6]$$

L'ensemble des points entiers que l'on peut obtenir comme combinaison linéaire rationnelle des vecteurs v1,v2,v3 est donc l'ensemble des combinaisons linéaires entières de w1, w2 et w3.

Par sécurité, vérifons que w est bien un tel vecteur.

```
> linsolve(augment(w1,w2,w3),w);
```

$$[1, 1, -1]$$

Ouf!

Exercice 3: réactions chimiques

Déterminer une base de l'ensemble des réactions chimiques possibles entre les molécules suivantes:

CH_4, CO, CO_2, O_2,H_2O, N_2, H_2, NH_3, N_2O_5.

```
> Molecules:=[CH4, CO, CO2, O2, H2O, N2, H2, NH3, N2O5];
```

$$Molecules := [CH4, CO, CO2, O2, H2O, N2, H2, NH3, N2O5]$$

```
> Mol:=[C+4*H, C+O, C+2*O, 2*O,2*H+O,2*N,2*H,N+3*H,
  2*N+5*O];
```

$$Mol := [C+4H, C+O, C+2O, 2O, 2H+O, 2N, 2H, N+3H, 2N+5O]$$

```
> At:=[C,H,O,N];
```

$$At := [C, H, O, N]$$

Une solution

Une réaction chimique faisant intervenir les molécules précédentes peut se voir comme une combinaison linéaire entière de ces molécules donnant 0.

```
> a:=vector(vectdim(Mol)):
  eq:=add(a[i]*Mol[i],i=1..9);
Error, (in vector) invalid arguments
```

$$eq := a_1 (C+4H) + a_2 (C+O) + a_3 (C+2O) + 2a_4 O + a_5 (2H+O) + 2a_6 N + 2a_7 H + a_8 (N+3H) + a_9 (2N+5O)$$

Les coefficients entiers a_i vérifient des équations linéaires car il y a conservation de la matière pour chaque atome!

```
> syst:=NULL:
  for x in At do
  syst:=syst,coeff(eq,x,1)=0:
  od:
  syst;
```

$$a_1+a_2+a_3=0,\ 4a_1+2a_5+2a_7+3a_8=0,\ a_2+2a_3+2a_4+a_5+5a_9=0,\ 2a_6+a_8+2a_9=0$$

Il ne reste quà trouver l'ensemble des solutions entières de ce système.

```
> M:=genmatrix({syst},[seq(a[i],i=1..9)]);
```

$$M := \mathrm{genmatrix}(\{a_1+a_2+a_3=0,\ 4a_1+2a_5+2a_7+3a_8=0,\ a_2+2a_3+2a_4+a_5+5a_9=0,\ 2a_6+a_8+2a_9=0\},\ [a_1, a_2, a_3, a_4, a_5, a_6, a_7, a_8, a_9])$$

```
> ismith(M,U,V);
```

Et en voici une base.

```
> seq(col(V,i),i=5..9);
```

[0, 2, -2, 1, 0, 0, 0, 0, 0], [-2, 5, -3, 0, 1, -1, 0, 2, 0], [-2, 4, -2, 0, 0, -1, 1, 2, 0], [-3, 6, -3, 0, 0, -2, 0, 4, 0], [0, 5, -5, 0, 0, -1, 0, 0, 1]

Ainsi par exemple, le premier vecteur correspond à la réaction chimique:

```
> dotprod(col(V,5),Molecules);
```

$$2\,CO - 2\,CO2 + O2$$

Un chimiste ne noterait pas ça tout à fait comme ceci.

Systèmes à paramètres

Exercice 1: rang d'une matrice à paramètres

Déterminer en fonctions des valeurs du paramètre α le rang de la matrice suivante:

```
> M:=matrix(6,6,(i,j)-> cos((2*i+j)*alpha));
```

$$M := \begin{bmatrix} \cos(3\,\alpha) & \cos(4\,\alpha) & \cos(5\,\alpha) & \cos(6\,\alpha) & \cos(7\,\alpha) & \cos(8\,\alpha) \\ \cos(5\,\alpha) & \cos(6\,\alpha) & \cos(7\,\alpha) & \cos(8\,\alpha) & \cos(9\,\alpha) & \cos(10\,\alpha) \\ \cos(7\,\alpha) & \cos(8\,\alpha) & \cos(9\,\alpha) & \cos(10\,\alpha) & \cos(11\,\alpha) & \cos(12\,\alpha) \\ \cos(9\,\alpha) & \cos(10\,\alpha) & \cos(11\,\alpha) & \cos(12\,\alpha) & \cos(13\,\alpha) & \cos(14\,\alpha) \\ \cos(11\,\alpha) & \cos(12\,\alpha) & \cos(13\,\alpha) & \cos(14\,\alpha) & \cos(15\,\alpha) & \cos(16\,\alpha) \\ \cos(13\,\alpha) & \cos(14\,\alpha) & \cos(15\,\alpha) & \cos(16\,\alpha) & \cos(17\,\alpha) & \cos(18\,\alpha) \end{bmatrix}$$

Comparer avec la réponse de Maple

```
> rank(M);
Error, (in linalg[gausselim]) unable to find a provably non-zero pivot
```

Une solution

Maple connaît bien ses formules de trigonométrie:

```
> expand(cos(3*alpha));
```

$$4\cos(\alpha)^3 - 3\cos(\alpha)$$

```
> N:=map(expand,M);
```

$N :=$

$[4\cos(\alpha)^3 - 3\cos(\alpha)\,,\, 8\cos(\alpha)^4 - 8\cos(\alpha)^2 + 1\,,$

$16\cos(\alpha)^5 - 20\cos(\alpha)^3 + 5\cos(\alpha)\,,\, 32\cos(\alpha)^6 - 48\cos(\alpha)^4 + 18\cos(\alpha)^2 - 1\,,$

$\%1\,,\, \%2]$

$[16\cos(\alpha)^5 - 20\cos(\alpha)^3 + 5\cos(\alpha)\,,\, 32\cos(\alpha)^6 - 48\cos(\alpha)^4 + 18\cos(\alpha)^2 - 1$

$,\, \%1\,,\, \%2\,,\, \%3\,,\, \%4]$

$[\%1\,,\, \%2\,,\, \%3\,,\, \%4\,,\, \%5\,,\, \%6]$

$[\%3\,,\, \%4\,,\, \%5\,,\, \%6\,,\, \%7\,,\, \%8]$

$[\%5\,,\, \%6\,,\, \%7\,,\, \%8\,,\, 16384\cos(\alpha)^{15} - 61440\cos(\alpha)^{13} + 92160\cos(\alpha)^{11}$

$- 70400\cos(\alpha)^9 + 28800\cos(\alpha)^7 - 6048\cos(\alpha)^5 + 560\cos(\alpha)^3 - 15\cos(\alpha)\,,$

$32768\cos(\alpha)^{16} - 131072\cos(\alpha)^{14} + 212992\cos(\alpha)^{12} - 180224\cos(\alpha)^{10}$

$+ 84480\cos(\alpha)^8 - 21504\cos(\alpha)^6 + 2688\cos(\alpha)^4 - 128\cos(\alpha)^2 + 1]$

$[\%7\,,\, \%8\,,\, 16384\cos(\alpha)^{15} - 61440\cos(\alpha)^{13} + 92160\cos(\alpha)^{11} - 70400\cos(\alpha)^9$

$+ 28800\cos(\alpha)^7 - 6048\cos(\alpha)^5 + 560\cos(\alpha)^3 - 15\cos(\alpha)\,,\, 32768\cos(\alpha)^{16}$

$- 131072\cos(\alpha)^{14} + 212992\cos(\alpha)^{12} - 180224\cos(\alpha)^{10} + 84480\cos(\alpha)^8$

$- 21504\cos(\alpha)^6 + 2688\cos(\alpha)^4 - 128\cos(\alpha)^2 + 1\,,\, 65536\cos(\alpha)^{17}$

$- 278528\cos(\alpha)^{15} + 487424\cos(\alpha)^{13} - 452608\cos(\alpha)^{11} + 239360\cos(\alpha)^9$

$- 71808\cos(\alpha)^7 + 11424\cos(\alpha)^5 - 816\cos(\alpha)^3 + 17\cos(\alpha)\,,\, 131072\cos(\alpha)^{18}$

$- 589824\cos(\alpha)^{16} + 1105920\cos(\alpha)^{14} - 1118208\cos(\alpha)^{12} + 658944\cos(\alpha)^{10}$

$- 228096 \cos(\alpha)^8 + 44352 \cos(\alpha)^6 - 4320 \cos(\alpha)^4 + 162 \cos(\alpha)^2 - 1]$

$\%1 := 64 \cos(\alpha)^7 - 112 \cos(\alpha)^5 + 56 \cos(\alpha)^3 - 7 \cos(\alpha)$

$\%2 := 128 \cos(\alpha)^8 - 256 \cos(\alpha)^6 + 160 \cos(\alpha)^4 - 32 \cos(\alpha)^2 + 1$

$\%3 := 256 \cos(\alpha)^9 - 576 \cos(\alpha)^7 + 432 \cos(\alpha)^5 - 120 \cos(\alpha)^3 + 9 \cos(\alpha)$

$\%4 :=$

$512 \cos(\alpha)^{10} - 1280 \cos(\alpha)^8 + 1120 \cos(\alpha)^6 - 400 \cos(\alpha)^4 + 50 \cos(\alpha)^2 - 1$

$\%5 := 1024 \cos(\alpha)^{11} - 2816 \cos(\alpha)^9 + 2816 \cos(\alpha)^7 - 1232 \cos(\alpha)^5$
$+ 220 \cos(\alpha)^3 - 11 \cos(\alpha)$

$\%6 := 2048 \cos(\alpha)^{12} - 6144 \cos(\alpha)^{10} + 6912 \cos(\alpha)^8 - 3584 \cos(\alpha)^6$
$+ 840 \cos(\alpha)^4 - 72 \cos(\alpha)^2 + 1$

$\%7 := 4096 \cos(\alpha)^{13} - 13312 \cos(\alpha)^{11} + 16640 \cos(\alpha)^9 - 9984 \cos(\alpha)^7$
$+ 2912 \cos(\alpha)^5 - 364 \cos(\alpha)^3 + 13 \cos(\alpha)$

$\%8 := 8192 \cos(\alpha)^{14} - 28672 \cos(\alpha)^{12} + 39424 \cos(\alpha)^{10} - 26880 \cos(\alpha)^8$
$+ 9408 \cos(\alpha)^6 - 1568 \cos(\alpha)^4 + 98 \cos(\alpha)^2 - 1$

Le résultat est horrible, mais tout de même: c'est une matrice polynômiale en $\cos(\alpha)$.
On peut donc lui appliquer l'algorithme de réduction!
Pour cela, mieux vaut tout d'abord poser $t = \cos(\alpha)$

```
> N1:=map2(subs,alpha=arccos(t),N):
  N2:=map(simplify,N1);
```

$N2 :=$

$[4 t^3 - 3 t, 8 t^4 - 8 t^2 + 1, 16 t^5 - 20 t^3 + 5 t, 32 t^6 - 48 t^4 + 18 t^2 - 1,$
$64 t^7 - 112 t^5 + 56 t^3 - 7 t, 128 t^8 - 256 t^6 + 160 t^4 - 32 t^2 + 1]$

$[16 t^5 - 20 t^3 + 5 t, 32 t^6 - 48 t^4 + 18 t^2 - 1, 64 t^7 - 112 t^5 + 56 t^3 - 7 t,$
$128 t^8 - 256 t^6 + 160 t^4 - 32 t^2 + 1, \%1, \%2]$

$[64 t^7 - 112 t^5 + 56 t^3 - 7 t, 128 t^8 - 256 t^6 + 160 t^4 - 32 t^2 + 1, \%1, \%2, \%3,$
$\%4]$

$[\%1, \%2, \%3, \%4, \%5, \%6]$

$[\%3, \%4, \%5, \%6,$
$16384 t^{15} - 61440 t^{13} + 92160 t^{11} - 70400 t^9 + 28800 t^7 - 6048 t^5 + 560 t^3 - 15 t,$
$32768 t^{16} - 131072 t^{14} + 212992 t^{12} - 180224 t^{10} + 84480 t^8 - 21504 t^6 + 2688 t^4$
$- 128 t^2 + 1]$

$[\%5, \%6,$
$16384 t^{15} - 61440 t^{13} + 92160 t^{11} - 70400 t^9 + 28800 t^7 - 6048 t^5 + 560 t^3 - 15 t,$
$32768 t^{16} - 131072 t^{14} + 212992 t^{12} - 180224 t^{10} + 84480 t^8 - 21504 t^6 + 2688 t^4$
$- 128 t^2 + 1, 65536 t^{17} - 278528 t^{15} + 487424 t^{13} - 452608 t^{11} + 239360 t^9$
$- 71808 t^7 + 11424 t^5 - 816 t^3 + 17 t, 131072 t^{18} - 589824 t^{16} + 1105920 t^{14}$
$- 1118208 t^{12} + 658944 t^{10} - 228096 t^8 + 44352 t^6 - 4320 t^4 + 162 t^2 - 1]$

$$\%1 := 256\,t^9 - 576\,t^7 + 432\,t^5 - 120\,t^3 + 9\,t$$

$$\%2 := 512\,t^{10} - 1280\,t^8 + 1120\,t^6 - 400\,t^4 + 50\,t^2 - 1$$

$$\%3 := 1024\,t^{11} - 2816\,t^9 + 2816\,t^7 - 1232\,t^5 + 220\,t^3 - 11\,t$$

$$\%4 := 2048\,t^{12} - 6144\,t^{10} + 6912\,t^8 - 3584\,t^6 + 840\,t^4 - 72\,t^2 + 1$$

$$\%5 := 4096\,t^{13} - 13312\,t^{11} + 16640\,t^9 - 9984\,t^7 + 2912\,t^5 - 364\,t^3 + 13\,t$$

$$\%6 := 8192\,t^{14} - 28672\,t^{12} + 39424\,t^{10} - 26880\,t^8 + 9408\,t^6 - 1568\,t^4 + 98\,t^2 - 1$$

```
> smith(N2,t);
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & -t+t^3 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

C'est finalement très simple: M est de rang 1 si $\cos(\alpha)^3 = \cos(\alpha)$ et de rang 2 sinon.

```
> solve(cos(alpha)^3 = cos(alpha));
```

$$\frac{1}{2}\pi,\ 0,\ \pi$$

Il est gratifiant de constater qu'on a résolu une question que Maple ne sait pas faire tout seul!

Exercice 2: systèmes d'équations polynomiales

Trouver l'ensemble des polynômes A, B et C en l'indéterminée x qui vérifient l'équation suivante:

A*P1+ B*P2+C*P3=1, où P1, P2 et P3 sont les polynômes suivants:

```
> P1:=(x^6-x^5+x^4-x^3+x^2-x+1)*x*(x-1);
```

$$P1 := (x^6 - x^5 + x^4 - x^3 + x^2 - x + 1)\,x\,(x-1)$$

```
> P2:=(x^8-x^6+x^4-x^2+1)*x*(x+1);
```

$$P2 := (x^8 - x^6 + x^4 - x^2 + 1)\,x\,(x+1)$$

```
> P3:=(x^12-x^11+x^9-x^8+x^6-x^4+x^3-x+1)*(x^2-1);
```

$$P3 := (x^{12} - x^{11} + x^9 - x^8 + x^6 - x^4 + x^3 - x + 1)\,(x^2-1)$$

Une solution

On applique la même méthode que pour les équations entières.

```
> M:=matrix(1,3,[P1,P2,P3]):
  smith(M,x,U,V);
```

$$\begin{bmatrix} 1 & 0 & 0 \end{bmatrix}$$

Une solution particulière est donné par

```
> part:=col(V,1);
```

$$part := \Big[-3\,x + \frac{3}{14}x^6 - \frac{9}{7}x^4 + \frac{3}{14}x^3 - 2\,x^2 + \frac{53}{14}x^8 - \frac{3}{14}x^{12} + 3\,x^{11} - x^9 - \frac{15}{14}$$

$$-\frac{20}{7}x^{10}+\frac{13}{14}x^{14}-\frac{25}{14}x^{13}, -\frac{1}{14}$$

$$(x^6-3x^5+4x^4-3x^3+x^2+1)(1-3x+23x^4+5x^2-21x^3+13x^6-25x^5), -1$$

$$]$$

```
> simplify(dotprod(part,[P1,P2,P3]));
```

$$1$$

L'ensemble de toutes les solutions est donné par:

```
> evalm(V &* vector([1,R,S]));
```

où R et S sont des paramètres à valeur polynomiale.

Là encore, Maple n'aurait pas su le faire tout seul!

ismith

Comparer le résultat de l'algorithme présenté en cours et celui de la fonction smith sur les matrices suivantes:

```
> M1:=diag(2,3,5);
  M2:=diag(2,3,4);
```

$$M1 := \begin{bmatrix} 2 & 0 & 0 \\ 0 & 3 & 0 \\ 0 & 0 & 5 \end{bmatrix}$$

$$M2 := \begin{bmatrix} 2 & 0 & 0 \\ 0 & 3 & 0 \\ 0 & 0 & 4 \end{bmatrix}$$

```
> N1:=ismith(M1,U1,V1);
  N2:=ismith(M2,U2,V2);
```

$$N1 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 30 \end{bmatrix}$$

$$N2 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 2 & 0 \\ 0 & 0 & 12 \end{bmatrix}$$

```
>
```

Voyez-vous une famille d'opérations élémentaires sur les lignes et les colonnes qui permette de passer de votre résultat à celui de Maple?
A quelle condition sur le vecteur entier B=vector([b1,b2,b3]) le système M1 &*X =B admet-il des solutions entières?
Comment reformuler ce résultat en utilisant la matrice N1?
Explication?

Une solution

Bizarre: les matrices M1 et M2 sont déjà diagonales et pourtant Maple les modifie!

Oui, l'algorithme de Smith de Maple est un peu plus complexe que l'algorithme présenté en cours.
Il a un avantage: le résultat a une forme normalisée, indépendante des choix que l'on peut être amené à faire au cours de l'algorithme. La forme normalisée de Maple est telle que les coefficients sur la diagonale se divisent successivement.

Pour passer de M1 à ismith(M1), on peut faire les opérations élémentaires suivantes:

```
> init(M1);
```

$$\begin{bmatrix} 2 & 0 & 0 \\ 0 & 3 & 0 \\ 0 & 0 & 5 \end{bmatrix}$$

```
> opC(2,-1,1);
```

$$\begin{bmatrix} 2 & -2 & 0 \\ 0 & 3 & 0 \\ 0 & 0 & 5 \end{bmatrix}$$

```
> opL(1,1,2);
```

$$\begin{bmatrix} 2 & 1 & 0 \\ 0 & 3 & 0 \\ 0 & 0 & 5 \end{bmatrix}$$

J'ai fait apparaître un 1.

```
> ech(1,1,1,2);
```

$$\begin{bmatrix} 1 & 2 & 0 \\ 3 & 0 & 0 \\ 0 & 0 & 5 \end{bmatrix}$$

```
> opC(2,-2,1);
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 3 & -6 & 0 \\ 0 & 0 & 5 \end{bmatrix}$$

```
> opL(2,-3,1);
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & -6 & 0 \\ 0 & 0 & 5 \end{bmatrix}$$

```
> opC(3,-1,2);
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & -6 & 6 \\ 0 & 0 & 5 \end{bmatrix}$$

```
> opL(2,-1,3);
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & -6 & 1 \\ 0 & 0 & 5 \end{bmatrix}$$

```
> ech(2,2,2,3);
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & -6 \\ 0 & 5 & 0 \end{bmatrix}$$

```
> opC(3,6,2);
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 5 & 30 \end{bmatrix}$$

```
> opL(3,-5,2);
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 30 \end{bmatrix}$$

Et voilà.

L'équation M1 &*X=B admet des solutions entières ssi b1 est pair, b2 est divisible par 3 et B3 est divisible par 5.
Si l'on utilise l'écriture U1M1V1=N1, on trouve la condition suivante:

```
> b:=vector(3):
  evalm(U1 &* b)[3];
```

$$15\, b_1 - 10\, b_2 + 6\, b_3$$

L'équation a des solutions entières ssi 15*b1-10*b2+6*b3 est divisible par 30.
Ces conditions sont équivalentes à celles données précédemment: c'est le théorème chinois mis à l'oeuvre!